1.7 Tečna a tečná rovina

```
> restart;
> with(LinearAlgebra):
> X:=Vector[row]([x,y,z,1]);
```

$$X := [x, y, z, 1]$$

Matice kvadriky:

```
> K:=Matrix(a,1..4,1..4,shape=symmetric);
```

$$K := \begin{bmatrix} a(1,1) & a(1,2) & a(1,3) & a(1,4) \\ a(1,2) & a(2,2) & a(2,3) & a(2,4) \\ a(1,3) & a(2,3) & a(3,3) & a(3,4) \\ a(1,4) & a(2,4) & a(3,4) & a(4,4) \end{bmatrix}$$

Rovnice kvadriky:

```
> Kv:=sort(expand(X.K.Transpose(X)),[x,y,z])=0;
```

$$Kv := a(1,1)\,x^2 + 2\,a(1,2)\,x\,y + 2\,a(1,3)\,x\,z + a(2,2)\,y^2 + 2\,a(2,3)\,y\,z + a(3,3)\,z^2 + 2\,a(1,4)\,x + 2\,a(2,4)\,y + 2\,a(3,4)\,z + a(4,4) = 0$$

Tecna kvadriky s bodem dotyku $M = [m, n, p]$:

```
> Tecna:=[x=m+t*u,y=n+t*v,z=p+t*w];
```

$$Tecna := [x = m + t\,u, y = n + t\,v, z = p + t\,w]$$

Rovnice průniku této tečny a kvadriky:

```
> Kv1:=simplify(eval(Kv,Tecna));
```

$$Kv1 := 2\,a(1,1)\,m\,t\,u + 2\,a(1,2)\,m\,t\,v + 2\,a(1,2)\,t\,u\,n + 2\,a(1,2)\,t^2\,u\,v + 2\,a(1,3)\,m\,t\,w + 2\,a(1,3)\,t\,u\,p + 2\,a(1,3)\,t^2\,u\,w + 2\,a(2,2)\,n\,t\,v + 2\,a(2,3)\,n\,t\,w + 2\,a(2,3)\,t\,v\,p + 2\,a(2,3)\,t^2\,v\,w + 2\,a(3,3)\,p\,t\,w + a(4,4) + 2\,a(2,4)\,n + 2\,a(3,4)\,p + a(1,1)\,m^2 + a(2,2)\,n^2 + a(3,3)\,p^2 + 2\,a(1,4)\,m + 2\,a(2,3)\,n\,p + a(1,1)\,t^2\,u^2 + 2\,a(1,2)\,m\,n + 2\,a(1,3)\,m\,p + a(2,2)\,t^2\,v^2 + a(3,3)\,t^2\,w^2 + 2\,a(1,4)\,t\,u + 2\,a(2,4)\,t\,v + 2\,a(3,4)\,t\,w = 0$$

Koeficienty A, B, C rovnice $At^2 + Bt + C = 0$ společných bodů kvadriky a tečny:

```
> A:=coeff(lhs(Kv1),t^2);
```

$$A := 2\,a(1,2)\,u\,v + 2\,a(1,3)\,u\,w + 2\,a(2,3)\,v\,w + a(1,1)\,u^2 + a(2,2)\,v^2 + a(3,3)\,w^2$$

```
> B:=1/2*coeff(lhs(Kv1),t);
```

$$B := a(1,1)\,m\,u + a(1,2)\,m\,v + a(1,2)\,u\,n + a(1,3)\,m\,w + a(1,3)\,u\,p + a(2,2)\,n\,v + a(2,3)\,n\,w + a(2,3)\,v\,p + a(3,3)\,p\,w + a(1,4)\,u + a(2,4)\,v + a(3,4)\,w$$

```
> C:=sort(coeff(lhs(Kv1),t,0),[m,n,p]);
```

$$C := a(1,1)\,m^2 + 2\,a(1,2)\,m\,n + 2\,a(1,3)\,m\,p + a(2,2)\,n^2 + 2\,a(2,3)\,n\,p + a(3,3)\,p^2 + 2\,a(1,4)\,m + 2\,a(2,4)\,n + 2\,a(3,4)\,p + a(4,4)$$

```
> B:=collect(B,[u,v,w]);
```

$$B := (a(1,1)\,m + a(1,3)\,p + a(1,4) + a(1,2)\,n)\,u + (a(1,2)\,m + a(2,2)\,n + a(2,4) + a(2,3)\,p)\,v + (a(3,3)\,p + a(2,3)\,n + a(1,3)\,m + a(3,4))\,w$$

Koeficienty rovnice tečné roviny:

```
> n1:=sort(coeff(B,u),[m,n,p]); n2:=sort(coeff(B,v),[m,n,p]);
  n3:=sort(coeff(B,w),[m,n,p]);
  n0:=sort(simplify(coeff(lhs(Kv1),t,0)-m*n1-n*n2-p*n3),[m,n,p]);
```

$$n1 := a(1,1)\,m + a(1,2)\,n + a(1,3)\,p + a(1,4)$$

$$n2 := a(1,2)\,m + a(2,2)\,n + a(2,3)\,p + a(2,4)$$

$$n3 := a(1,3)\,m + a(2,3)\,n + a(3,3)\,p + a(3,4)$$

$$n0 := a(1,4)\,m + a(2,4)\,n + a(3,4)\,p + a(4,4)$$

Tečná rovina τ:

```
> tau:=n1*u+n2*v+n3*w+n0=0;
```

$$\tau := (a(1,1)\,m + a(1,2)\,n + a(1,3)\,p + a(1,4))\,u + (a(1,2)\,m + a(2,2)\,n + a(2,3)\,p + a(2,4))\,v + (a(1,3)\,m + a(2,3)\,n + a(3,3)\,p + a(3,4))\,w + a(1,4)\,m + a(2,4)\,n + a(3,4)\,p + a(4,4) = 0$$

Můžeme ji psát v maticovém tvaru:

```
> tau:=collect(evalm(Vector[row]([m,n,p,1]).K.Transpose(X)),[x,y,z
  ])=0;
```

$$\tau := x\,(a(1,1)\,m + a(1,2)\,n + a(1,3)\,p + a(1,4)) + y\,(a(1,2)\,m + a(2,2)\,n + a(2,3)\,p + a(2,4)) + z\,(a(1,3)\,m + a(2,3)\,n + a(3,3)\,p + a(3,4)) + a(1,4)\,m + a(2,4)\,n + a(3,4)\,p + a(4,4) = 0$$

Příklad: Určete, při které hodnotě k se rovina $x - 2y - 2z + k = 0$ dotýká kvadriky $x^2 + 4y^2 + 16z^2 - 144 = 0$ (Str. 22).

```
> restart;
> with(LinearAlgebra):
> K:=Matrix(a,1..4,1..4,shape=symmetric):
> X:=Vector[row]([x,y,z,1]):
> rho:=x-2*y-2*z+k=0;
```

$$\rho := x - 2y - 2z + k = 0$$

```
> kv:=x^2+4*y^2+16*z^2-144=0;
```

$$kv := x^2 + 4y^2 + 16z^2 - 144 = 0$$

Matice kvadriky:

```
> a(1,1):=coeff(lhs(kv),x^2); a(2,2):=coeff(lhs(kv),y^2);
```

```
a(3,3):=coeff(lhs(kv),z^2);
a(1,2):=1/2*coeff(coeff(lhs(kv),x),y);
a(1,3):=1/2*coeff(coeff(lhs(kv),x),z);
a(1,4):=1/2*coeff(coeff(coeff(lhs(kv),x,1),y,0),z,0);
a(2,3):=1/2*coeff(coeff(lhs(kv),y,1),z);
a(2,4):=1/2*coeff(coeff(coeff(lhs(kv),y,1),x,0),z,0);
a(3,4):=1/2*coeff(coeff(coeff(lhs(kv),z,1),y,0),x,0);
a(4,4):=coeff(coeff(coeff(lhs(kv),x,0),y,0),z,0);
```

$$a(1,1) := 1$$
$$a(2,2) := 4$$
$$a(3,3) := 16$$
$$a(1,2) := 0$$
$$a(1,3) := 0$$
$$a(1,4) := 0$$
$$a(2,3) := 0$$
$$a(2,4) := 0$$
$$a(3,4) := 0$$
$$a(4,4) := -144$$

```
> K;
```

$$\begin{bmatrix} 1 & 0 & 0 & 0 \\ 0 & 4 & 0 & 0 \\ 0 & 0 & 16 & 0 \\ 0 & 0 & 0 & -144 \end{bmatrix}$$

Tečná rovina v bodě $M = [m, n, p]$:

```
> tau1:=collect(Vector[row]([m,n,p,1]).K.Transpose(X),[x,y,z])=0;
```

$$\tau 1 := -144 + x\,m + 4\,y\,n + 16\,z\,p = 0$$

```
> tau1:=expand(tau1/m);
```

$$\tau 1 := -\frac{144}{m} + x + \frac{4\,y\,n}{m} + \frac{16\,z\,p}{m} = 0$$

```
> coeffs(lhs(tau1),[x,y,z])-coeffs(lhs(rho),[x,y,z]);
```

$$-\frac{144}{m} - k,\ 0,\ \frac{4\,n}{m} + 2,\ \frac{16\,p}{m} + 2$$

```
> res:=solve({coeffs(lhs(tau1),[x,y,z])-coeffs(lhs(rho),[x,y,z])},
  {m,n,p});
```

$$res := \{m = -\frac{144}{k},\ n = \frac{72}{k},\ p = \frac{18}{k}\}$$

```
> kv1:=eval(kv,[x=m,y=n,z=p]);
```

$$kv1 := m^2 + 4\,n^2 + 16\,p^2 - 144 = 0$$

```
> kv2:=eval(kv1,res);
```

$$kv2 := \frac{46656}{k^2} - 144 = 0$$

Úloha má řešení pro dvě hodnoty k:

```
> k_res:=solve(kv2,{k});
```

$$k\_res := \{k = -18\}, \{k = 18\}$$

```
> assign(k_res[1]);
> k; M1:=eval([m,n,p],res); tau2[1]:=eval(tau1,res);
```

$$-18$$

$$M1 := [8, -4, -1]$$

$$\tau 2_1 := -18 + x - 2y - 2z = 0$$

```
> k:='k';
```

$$k := k$$

```
> assign(k_res[2]);
> k; M2:=eval([m,n,p],res); tau2[2]:=eval(tau1,res);
```

$$18$$

$$M2 := [-8, 4, 1]$$

$$\tau 2_2 := 18 + x - 2y - 2z = 0$$

```
> tr1:=solve(tau2[1],z); tr2:=solve(tau2[2],z);
```

$$tr1 := -9 + \frac{x}{2} - y$$

$$tr2 := 9 + \frac{x}{2} - y$$

```
> plotsetup(inline,plotoptions=`portrait,noborder,shrinkby=0`);
> kvg:=plots[implicitplot3d](kv,x=-15..15,y=-10..10,z=-8..8,grid=[
  30,30,30],style=patchcontour,color=red,lightmodel=light1,scaling
  =constrained):
> M1g:=plottools[sphere](M1,0.3):
> M2g:=plottools[sphere](M2,0.3):
> tau1g:=plots[contourplot3d](tr1,x=-2..18,y=-10..8,grid=[2,2],con
  tours=40,color=COLOR(RGB,100/255,100/255,100/255),filled=false):
> tau2g:=plots[contourplot3d](tr2,x=-18..2,y=-6..12,grid=[2,2],con
  tours=40,color=COLOR(RGB,130/255,130/255,130/255),filled=false):
> plots[display](kvg,tau1g,tau2g,M1g,M2g);
```



>

>